



PharmaSUG 2024 - Paper DS- 374

Implementation of Composite Estimands for Responder Analysis based on Change from Baseline in Non-solid Tumours

Reshma Radhakrishnan, Efficacy Lifescience Analytics

ABSTRACT

The International Council for Harmonization of Technical Requirements for Pharmaceuticals for Human Use (ICH) published E9 R1 “Addendum on Estimands and Sensitivity Analysis in Clinical Trials” in August 2017. An estimand is a precise description of the treatment effect reflecting the clinical question posed by a given clinical trial objective. Description of estimands must be based on how intercurrent events affect clinical question of interest. Intercurrent events (ICE) are events occurring after treatment start and how it affects either the interpretation or measurement associated with clinical question of interest. Some of examples of ICEs are Death, switch to other treatment, treatment discontinuation etc. There are five strategies for addressing ICEs like Treatment policy strategy, Hypothetical strategies, Composite variable strategies, While on treatment strategies and Principal stratum strategies. After regulatory authorities endorsed it from 2020, Estimands gained more popularity and included in Statistical Analysis Plan (SAP). It would be helpful to understand estimands and, how efficiently one can construct data sets to support efficacy analysis to make sure correct implementation of ICEs.

This paper aims to provide its audience the simplified way to design data sets, variables for composite estimands for responder analysis based on change from baseline. Also, how to carefully handle ICEs in missing data analysis using SAS procedures PROC STDRATE and PROC MI. The STDRATE procedure computes directly standardized rates and risks for study populations using Mantel-Haenszel estimates and computes indirectly standardized rates and risks, including SMR whereas PROC MI is used to replace missing values with multiple imputation.

INTRODUCTION

A systematic and well-defined approach in defining clinical question of interest will improve the way how we plan and conduct clinical trial and, avoid misinterpretation of treatment effect. So, Estimands and its framework are introduced to address this gap by ICH. Estimands are the parameters to be estimated from clinical trial. The choice of strategies can influence how more conventional attributes of a trial are reflected when describing the clinical question, for example the treatments, population, or the variable (endpoint) of interest. The statistical analysis of clinical trial data should be aligned to the estimand. Estimand framework will have detailed methodology and strategy of how each of the ICEs will be handled and affect interpretation of endpoints of clinical interest. Intercurrent Events are events which occurs after treatment initiation and will have impact on clinical question of interest.

The analysis of estimand may additionally be challenging due to missing data. It is important to understand that missing data and ICE are not same. Missing data are the data that were not collected for an estimator which is different from data that are not considered meaningful because of an ICE. Missing data that are not collected due to lost to follow up or any other reason will not be necessarily considered as Intercurrent Events.

FRAMEWORK

This framework enables proper trial planning that clearly distinguishes between the target of estimation (trial objective, estimand), the method of estimation (estimator), the numerical result and a sensitivity analysis.

ESTIMAND ATTRIBUTES

1. **Population:** Patients targeted by the clinical question.
2. **Treatment:** Treatment of interest and comparative treatment.
3. **Variable (or endpoint):** Parameter of patients that is required to address the clinical question.
4. **Intercurrent Event:** List of ICEs which will affect clinical question and strategy used to address it.
5. **Summary Measure:** Summary statistics used for treatment comparison.


STRATEGIES FOR ADDRESSING INTERCURRENT EVENTS

1. **Treatment Policy Strategy:** Variable value will be considered despite of the ICE occurrence.
2. **Hypothetical Strategy:** Assumes ICE would not occur to patients. Variable value after ICE would be statistically modelled (multiple imputation or other missing data handling) and used for analysis.
3. **Composite Strategy:** Outcome of variable value will be affected based on the occurrence of ICE.
4. **While on Treatment Strategy:** Variable value prior to ICE occurrence will be considered.
5. **Principal Stratum Strategy:** Variable value will be considered only for subset of patients from targeted population.

COMPOSITE STRATEGY

Composite strategy will incorporate intercurrent events into the endpoint. A composite variable strategy can avoid statistical assumptions about data after an intercurrent event by considering occurrence of the intercurrent event as a component of the outcome. Examples of composite strategy: Treatment discontinuation, Death, Treatment switch, Use of rescue medication.

Figure 1: Illustration of how ICEs addressed using composite strategy for an endpoint.



MAIN ESTIMAND FRAMEWORK

Let us consider a phase III trial, randomized, placebo-controlled trial evaluating safety and efficacy of active treatment in patients with high-risk myelofibrosis.

The primary endpoint is Spleen Volume Response Rate (SVRR), the proportion of patients achieving $\geq 35\%$ spleen volume reduction from baseline on MRI/CT at End of Cycle 6 (EOC6).

Let us consider below the estimand framework for SVRR primary endpoint.

Population	All randomized subjects	
Treatment	Six cycles (28 days if each cycle) of active treatment or Standard of Care (SOC)	
Variable (endpoint)	Participants who achieve an $\leq 35\%$ spleen volume response rate at end of cycle 6	
Intercurrent Events (ICEs)	Event	Strategy
	Participants who died before EOC6 assessment	Composite Strategy. Considered as non-responder.
	Participants who switched treatment before EOC6 assessment	Composite Strategy. Considered as non-responder.
	Participants who discontinued treatment before EOC6 assessment	Composite Strategy. Considered as non-responder.
Summary measure	Difference in proportion between Active Treatment and SOC at EOC6	

DESIGN VARIABLES IN DATA SET

PREREQUISITES

- BDS data set ADSN which will have Spleen Volume Assessment and change from baseline for each cycle.
- Estimand framework from SAP.

STEP 1

We need to identify the list of ICEs from estimand framework. So, there are three ICEs defined based on the above estimand framework for which we need three variables.

1. ICDTH – Flag ‘Y’ for subjects who died before EOC6 assessment.
2. ICSWTH – Flag ‘Y’ for subjects who switched treatment before EOC6 assessment.
3. ICDISC – Flag ‘Y’ for subjects who discontinued treatment before EOC6 assessment.

STEP 2

Identifying where we are going to create these ICE variables. The best practice would be to create in subject level data set ADSL instead of BDS data set ADSN or any intermediate data set in each of efficacy data set.

Let us look further into the reason about the recommended structure to keep ICE variables in subject level data set ADSL.

1. These ICE variables are subject level data rather than record level data.
2. When we perform missing data analysis for change from baseline for spleen volume response using multiple imputation method in table level. We need to re-create response variable (Y/N) based on new data set and need to use our ICEs to determine whether that subject is responder or not. So, if we have our ICEs in intermediate data set or in BDS data set, we need to rederive this variable in tables programming which will be cumbersome and less efficient.

MULTIPLE IMPUTATION FOR SVRR

PREREQUISITES

- ADSL (including three ICE variables)
- ADSN (Spleen Volume Assessment at each visit and respective change from baseline).

In this section, we are going to create report programming for the primary endpoint - spleen volume response rate (SVRR), the proportion of patients achieving $\geq 35\%$ spleen volume reduction from baseline on MRI/CT at EOC6 using multiple imputation PROC MI and PROC STD RATE.

STEP 1

Simulated Spleen volume assessment for each cycle and ADSL are created as per source code in appendix based. Let us start from intermediate data set TDADSN_ICE which will have three ICE variables from ADSL, stratification variable STRAT1, baseline spleen volume assessment SPVL1 and change from baseline of cycle 3 CHG3 and Cycle 6 CHG6 will be present. This will have one record per subject data.

USUBJID	TRTPN	ICDTH	ICSWTH	ICDISC	STRAT1	SPVL1	CHG3	CHG6
101-001	1	Y	N	N	1	407	-57	44
101-006	1	N	N	N	1	219	.	21
101-009	1	Y	Y	N	1	313	0	-113
101-012	1	Y	N	N	1	616	-509	-405
101-016	1	Y	Y	N	1	370	-61	62
101-032	1	Y	N	N	1	216	.	187
101-034	1	N	Y	N	1	287	-118	-234
101-036	1	N	Y	N	1	295	-178	68
101-038	1	Y	N	Y	1	154	107	208
101-041	1	N	N	Y	1	522	-105	-137
101-042	1	N	Y	Y	1	350	-5	76
101-043	1	N	Y	N	1	428	-218	-18

STEP 2

We need to check missing data pattern is monotone or arbitrary. In this paper, we will concentrate on arbitrary missing data pattern for continuous variable CHG3 and CHG6.

```
proc mi data=tdadsn_ice nimpute=0;
  class trtpn strat1;
  fcs;
  var trtpn strat1 spvl1 chg3 chg6;
run;
```

Missing Data Patterns										
Group	TRTPN	STRAT1	SPVL1	CHG3	CHG6	Freq	Percent	Group Means		
								SPVL1	CHG3	CHG6
1	X	X	X	X	X	163	81.50	302.061350	20.865031	-8.404908
2	X	X	X	X	.	20	10.00	288.800000	16.500000	.
3	X	X	X	.	X	16	8.00	295.437500	.	-18.437500
4	X	X	X	.	.	1	0.50	345.000000	.	.

From the above result, we can understand that CHG3 and CHG6 have missing data arbitrary.

STEP 3

We are creating 30 multiple imputation data set by FCS REG regression method for CHG3 and CHG6. Please note the order of variable in FCS statement is important.

```
proc mi data=tdadsn_ice out=tdadsn_imp nimpute=30 seed=151121
round=. . . .;
  class strat1;
  var strat1 spvl1 chg3 chg6;
  by trtpn;
  fcs nbiter=200 reg(chg3=spvl1 strat1);
  fcs nbiter=200 reg(chg6=spvl1 chg3 strat1);
run;
```

STEP 4

Now, we need to derive percent change from baseline. If the percentage change from baseline shows 35% reduction in spleen size, then those subjects are responders. Others are non- responders.

Here is the key step for incorporating composite ICEs to identify whether subjects become non- responder or not. If any of our responder flags are Y, then we need to flip that subject to “Non – responder.”

```
data tdadsn_yn;
  set tdadsn_imp;
  pchg6=chg6/spvl1*100;
  if pchg6<=-35 then
    svr35fln=1;
  else
    svr35fln=0;
  if icdth='Y' or icswth='Y' or icdisc='Y' then
    svr35fln=0;
run;
```

STEP 6

Then we can use the above data set to create point estimate, p value or difference in proportion using the CMH model adjusted by the stratification factor. PROC STDRAE will be used for this analysis. _imputation_ variable specifies number of iterations.

```
proc stdrate data=tdsum_1 method=mh stat=risk effect=diff;
  by _imputation_;
  population group=trtpn event=resp_cnt total=tot_group_cnt;
  strata strat1 /order=data stats effect;
  ods output stdrisk=tdrisk_svrr effect=tdeffectout_svrr;
run;
```

STEP 7

As a last step, we need to combine the above result from separate datasets using PROC MIANALYZE to derive standard rates or risk difference as per study requirement.

```
proc mianalyze data=tdeffectout_svrr;
  modeleffects riskdiff;
  stderr stderr;
  ods output parameterestimates=tdrisk_svrr;
run;
```

CONCLUSION

Estimand Framework and implementation of ICEs are new concepts which are slowly becoming mandatory information in Statistical Analysis Plan (SAP) and may be quite overwhelming for programmers. In this paper, we presented estimand, handling intercurrent events using composite strategy, how to implement ICEs in responder analysis for SVRR using multiple imputation method. As a programmer, care should be taken to understand data structure with ICEs for responder analysis that includes multiple imputation.

REFERENCES

Chris Smith, Scott Kosten, 2023, PharmaSUG 2017- “Multiple Imputation: A Statistical Programming Story”. - [Multiple Imputation: A Statistical Programming Story \(pharmasug.org\)](https://pharmasug.org)

ICH E9(R1) – “Addendum on Estimands and Sensitivity Analysis in Clinical Trials”

https://database.ich.org/sites/default/files/E9-R1_Step4_Guideline_2019_1203.pdf

ICH E9(R1) Expert Working Group, June 2018, “Draft (Step 2) guideline ICH E9(R1)”

https://database.ich.org/sites/default/files/E9-R1_EWG_Step2_TrainingMaterial.pdf

ACKNOWLEDGMENTS

My sincere thanks to Efficacy Lifescience Analytics for giving me an opportunity to author this paper and Mritunjay Kumar my manager for providing motivation and support for this paper. Any brand and product names are trademarks of their respective companies.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact author at:

Reshma Radhakrishnan, Principal Statistical Programmer - Biostatistics & Programming
 Efficacy Lifescience Analytics
 Bengaluru, India
 E-mail: reshma.radhakrishnan@ephicacy.com
www.ephicacy.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

APPENDIX

```
/******  
/* source code for responder analysis for SVRR using ICE as composite strategy  
*/  
/******  
  
proc format;  
    value vis 1='b1' 3='EOC3' 6='EOC6';  
run;  
  
/* step1: create adsl and spleen volume simulation dataset */  
data adsl;  
    length usubjid $7.;  
    call streaminit(1977);  
  
    do id=1 to 200;  
        usubjid='101-'||put(id, z3.);  
        *assign treatment group 1=active 2=soc;  
  
        if rand('bernoulli', 0.5)=1 then  
            trtpn=1;  
        else  
            trtpn=2;  
        *ICE subjects who died before EOC6;  
  
        if rand('bernoulli', 0.5) then  
            icdth='Y';  
        else  
            icdth='N';  
        *ICE subjects who switched treatment before EOC6;  
  
        if rand('bernoulli', 0.5) then  
            icswth='Y';  
        else  
            icswth='N';  
        *ICE subjects who discontinued treatment before EOC6;  
  
        if rand('bernoulli', 0.5) then  
            icdisc='Y';  
        else  
            icdisc='N';  
        *Population flag;  
  
        if rand('bernoulli', 0.05) then  
            ittfl='Y';  
        else  
            ittfl='N';  
        output;
```

```

        end;
        drop id;
run;

data dummy;
    length usubjid $7. paramcd $8. param $15.;
    *visit structure;
    array visits (3) _temporary_ (1 3 6);
    *sets seed for all rand functions ;
    call streaminit(1977);
    *data simulation loop;

    do id=1 to 200;
        *assign subject id;
        usubjid='101-'||put(id, z3.);
        *assign stratification 1= high risk 2= intermediate risk;

        if rand('bernoulli', 0.5)=1 then
            strat1=1;
        else
            strat1=2;
        *assign treatment group 1=active 2=soc;

        if rand('bernoulli', 0.5)=1 then
            trtpn=1;
        else
            trtpn=2;
        paramcd='SPVOL';
        param='Spleen Volume';
        *simulate results;

        do i=1 to dim(visits);
            acyclen=visits(i);
            acycle=put(acyclen, vis.);

            if trtpn=1 then
                do;
                    if visits(i)=1 then
                        aval=round(rand('normal', 300, 110));
                    else
                        aval=round(rand('normal', 300, 110)+visits(i));
                    end;
                else
                    do;
                        aval=round(rand('normal', 300, 110));
                    end;
                output;
            end;
        end;
        drop i id;
run;

/* simulate missing data; */
data adsn;
    set dummy;
    by usubjid;
    call streaminit(1980);

```



```

        *simulate arbitrary missing data pattern;

        if acycle ne 'b1' then
            do;

                if rand('bernoulli', 0.08) then
                    delete;

            end;
run;

/* step 2: derive change from baseline*/
proc transpose data=adsn out=tdadsn_trans prefix=spvl;
    by usubjid trtpn strat1 paramcd param;
    id acyclen;
    var aval;
run;

data tdadsn_chg;
    set tdadsn_trans;

    if not missing(spvl1) and not missing(spvl3) then
        chg3=spvl3-spvl1;

    if not missing(spvl1) and not missing(spvl6) then
        chg6=spvl6-spvl1;
    drop _name_;
run;

/* step 3: merge adsl and spleen volume dataset */
proc sort data=tdadsn_chg;
    by usubjid;
run;

proc sort data=adsl;
    by usubjid;
run;

data tdadsn_ice;
    merge adsl(in=a) tdadsn_chg(in=b);
    by usubjid;

    if a;
run;

/* step4 : check missing data pattern - stratal should not be missing*/
proc mi data=tdadsn_ice nimpute=0;
    class trtpn strat1;
    fcs;
    var trtpn strat1 spvl1 chg3 chg6;
run;

/* arbitrary missing data pattern observed. use fcs reg in proc mi*/
proc sort data=tdadsn_ice out=tdadsn_ice;
    by trtpn strat1 usubjid;
run;

```

```

/* multiple imputation */
proc mi data=tdadsn_ice out=tdadsn_imp nimpute=30 seed=151121 round=. . . .;
  class strat1;
  var strat1 spvl1 chg3 chg6;
  by trtpn;
  fcs nbiter=200 reg(chg3=spvl1 strat1);
  fcs nbiter=200 reg(chg6=spvl1 chg3 strat1);
run;

/* step 5: derive percent change from baseline and use ices to create binary
endpoint*/
data tdadsn_yn;
  set tdadsn_imp;
  pchg6=chg6/spvl1*100;
  * primary endpoint: ≥35% SVRR from baseline on mri/ct at EOC6 ;
  * 1=responder, 0=non-responder;

  if pchg6<=-35 then
    svr35fln=1;
  else
    svr35fln=0;
  * implement composite variable strategy - any of ICE flag is y,
  then they will be non-responder;

  if icdth='Y' or icswth='Y' or icdisc='Y' then
    svr35fln=0;
run;

/* step 6: stratified analysis of SVRR using proc stdrate */
/* resp_cnt - number of subjects who are responder or non-responder in a group
non-responder count will always be zero */
/* tot_group_count - group count */
proc summary data=tdadsn_yn;
  class _imputation_ strat1 trtpn;
  output out=tdsum sum(svr35fln)=resp_cnt n(svr35fln)=tot_group_cnt;
run;

data tdsum_1;
  set tdsum;
  where not missing(_imputation_) and not missing(strat1) and not
missing(trtpn);
run;

proc sort;
  by _imputation_;
run;

/* calculate risk difference for each dataset */
proc stdrate data=tdsum_1 method=mh stat=risk effect=diff;
  by _imputation_;
  population group=trtpn event=resp_cnt total=tot_group_cnt;
  strata strat1 /order=data stats effect;
  ods output stdrisk=tdrisk_svrr effect=tdeffectout_svrr;
run;

```

```

/* combine results from multiple imputed dataset */
proc sort data=tdrisk_svrr;
    by trtpn;
run;

proc mianalyze data=tdeffectout_svrr;
    modeleffects riskdiff;
    stderr stderr;
    ods output parameterestimates=tdrisk_svrr;
run;

```